Simple ejector pump analysis / Une analyse simple d'une pompe à éjection

Turyk, P. J.

Publisher's version / Version de l'éditeur:
https://doi.org/10.4224/40003617
Laboratory Memorandum (National Research Council Canada. Division of Mechanical Engineering. Engine Laboratory); no. LM-ENG-014, 1988-08

CONTROLLED
UNCLASSIFIED

CONTRÔLÉE
NON CLASSIFIÉE

Laboratory Memorandum

Mémoire de laboratoire

1988 / 08

LM-ENG-014

SIMPLE EJECTOR PUMP ANALYSIS

P.J. Turyk

Division of
Mechanical Engineering

Division de
génie mécanique

National Research
Council Canada

Conseil national
de recherches Canada

Canada

DIVISION OF MECHANICAL ENGINEERING PUBLICATIONS

DM **(Division of Mechanical Engineering Report)**
Scientific and technical information considered important, complete and a lasting contribution to existing knowledge.

TR **(Technical Report)**
Information less broad in scope but a substantial contribution to existing knowledge.

CTR **(Controlled/Classified Technical Report)**
A Technical Report with controlled distribution for national security, proprietary or other reasons.

LM **(Laboratory Memorandum)**
Preliminary or exploratory information with controlled distribution.

CAT **(Calibration Analysis and Test Report)**
Information on minor laboratory projects or services.

PUBLICATIONS DE LA DIVISION DE GÉNIE MÉCANIQUE

DM **(Rapport de la Division de génie mécanique)**
Informations scientifiques et techniques jugées importantes, complètes et susceptibles de contribuer de façon durable à l'avancement des connaissances courantes.

TR **(Rapport technique)**
Informations de moindre importance, mais pouvant contribuer substantiellement à l'avancement des connaissances actuelles.

CTR **(Rapport technique à diffusion contrôlé/classifié)**
Rapport technique à diffusion contrôlée pour des raisons de sécurité nationale, de propriété intellectuelle et autres.

LM **(Mémoire de laboratoire)**
Informations préliminaires ou de nature exploratoire à diffusion contrôlée.

CAT **(Rapport d'étalonnage d'analyse et d'essai)**
Informations sur de petits projets ou des services de laboratoire.

SIMPLE EJECTOR PUMP ANALYSIS

UNE ANALYSE SIMPLE D'UNE POMPE A ÉJECTION

P.J. Turyk

Laboratory Memorandum

Mémoire de laboratoire

1988/08

LM-ENG-014

Engine Laboratory
Laboratoire des moteurs

ABSTRACT

A simple one-dimensional solution of the conservation of mass, momentum, energy, and equation of state for perfect gases for an ejector pump is presented. The report includes the mathematical formulation of the solution and a sample calculation. A computer program is presented in the Appendices.

ABSTRAIT

Ce rapport présente une solution simple à une-dimension à la conservation de masse, la force d'impulsion, l'énergie, et l'équation pour du gas parfait pour une pompe à éjection. Le rapport inclus la formule mathématique de la solution ainsi qu'un example de calcul. Un programme d'ordinateur est présenté dans les appendices.

CONTENTS

LIST OF ILLUSTRATIONS

LIST OF TABLES

LIST OF APPENDICES

LIST OF SYMBOLS

Symbol		
A	area	(m^2)
b	y intercept of y=mx+b	
c_p	specific heat at constant pressure	(Btu/lbm/°R; J/kg/K)
D	diameter	(in; m)
h	enthalpy	(Btu/lbm; J/kg)
K	defined in equations (16)	
m	mass flow, slope (Appendix B)	(lbm/min; kg/s)
P	pressure	(psi; Pa)
ρ	density	(lbm/ft^3; kg/m^3)
Q	volume flow	(cfm; m^3/s)
R	gas constant	(Btu/lbm/°R; J/kg/K)
S	standoff distance	(in; m)
T	temperature	(°R, K)
u	gas velocity	(ft/min; m/s)
x	independent variable (Appendix B)	
y	dependent variable (Appendix B)	

Subscripts

amb	ambient
calc,c	calculated value
guess,g	guessed value
m	mixing tube
1,2,3	station numbers; indices for variable K
4,5,6	indices for variable K

Superscripts

'	subindex for variable K
1,2,3	iteration numbers (Appendix B)

SIMPLE EJECTOR PUMP ANALYSIS

1.0 INTRODUCTION

The following Laboratory Memorandum presents a simple analysis of the fluid mechanics of an ejector pump type of device. The work presented in this memorandum was performed in response to a request from Capt. B. Cook (1987) of the Aerospace Maintenance and Development Unit (AMDU), DND, CFB Trenton.

Ejector pumps, or eductors, are used to induce a flow of secondary fluid by means of a high velocity primary stream of fluid. The personnel at AMDU wished to evaluate the feasibility of utilizing the ejector pump principle, using air from a compressed air start unit (CSU) as the primary stream, to entrain a certain amount of secondary air from the atmosphere. The total airflow would be used for the ground testing of a ram air turbine-powered generator.

The purpose of the analysis presented in this report was to estimate the amount of total air which would be entrained in such an eductor arrangement. AMDU personnel would then be able to determine whether an experimental program is warranted. The simple analysis presented here would save the costs of setting up such a program in the event that the analysis showed that the proposed eductor arrangement would not entrain sufficient air to ground test the ram air turbine.

The analysis presented in this report is limited to a simple one-dimensional solution of the conservation of mass, momentum, energy, and state equation for a perfect gas for an ejector pump. A complete list of assumptions are listed in the text. The report includes the mathematical formulation of the solution, a sample calculation using input from the AMDU request, and concluding comments. A FORTRAN computer program called SEPP which performs the calculations is presented in the Appendices.

2.0 SIMPLE ENGINEERING ANALYSIS OF AN EJECTOR PUMP SYSTEM

Given the geometry, primary airflow and ambient conditions of an ejector pump (eductor) system, it is desired to calculate the induced secondary airflow and the exit airflow conditions. The geometry is shown in Figure 1. The known parameters in the system are:

- the geometry: D_1, D_3
- inlet flow conditions: m_1, or u_1, T_1
- secondary airflow temperature: T_2
- ambient conditions: P_{amb}, T_{amb}

The following assumptions are made:

- circular geometry
- one-dimensional, uniform flow at all stations
- perfect mixing of secondary and primary airstreams
- no turbulence
- inviscid flow
- no momentum loss factors
- no secondary airflow entry loss factors
- no friction in mixing tube
- no momentum or kinetic energy correction factors
- far from the ejector, $u_{amb}=0$
- uniform radial pressure distribution, i.e. $P_1 = P_2$
- ISA/SLS ambient conditions (P_{amb} = 101325 Pa, T_{amb} = 15°C)
- $P_3 = P_{amb}$

In addition, if the mass flow m_1 is known, u_1 can be estimated from:

$$m = \rho A u \tag{1}$$

where

$$P = \rho R T \tag{2}$$

and P_1 can be taken as atmospheric, with little loss in accuracy.

It is necessary here to calculate u_2, u_3, P_1, and T_3. These parameters can be found by solving the conservation of mass, momentum, and energy equations, in addition to the equations of state for air:

$$\text{mass:} \quad m_1 + m_2 = m_3 \tag{3}$$

$$\text{momentum:} \quad m_1u_1 + P_1A_1 + m_2u_2 + P_2A_2 = m_3u_3 + P_3A_3 \tag{4}$$

energy: $$m_1\left[\frac{u_1^2}{2} + h_1\right] + m_2\left[\frac{u_2^2}{2} + h_2\right] = m_3\left[\frac{u_3^2}{2} + h_3\right] \quad (5)$$

state: $$\Delta h = c_p \, \Delta T \quad (6)$$

The equation for mass flow, (1), and the perfect gas relation, (2), are also used.

When equations (1), (2), and (6) are substituted into equations (3), (4), and (5), and the assumption that $P_1 = P_2$ is used, the results are:

$$P_1\left[\frac{A_1u_1}{RT_1} + \frac{A_2u_2}{RT_2}\right] = \frac{P_3A_3u_3}{RT_3} \quad (7)$$

$$P_1\left[A_3 + \frac{A_1u_1^2}{RT_1} + \frac{A_2u_2^2}{RT_2}\right] = P_3A_3\left[1 + \frac{u_3^2}{RT_3}\right] \quad (8)$$

$$P_1\left[\frac{A_1u_1}{RT_1}\left[c_pT_1 + \frac{u_1^2}{2}\right] + \frac{A_2u_2}{RT_2}\left[c_pT_2 + \frac{u_2^2}{2}\right]\right] = \frac{P_3A_3u_3}{RT_3}\left[c_pT_3 + \frac{u_3^2}{2}\right] \quad (9)$$

Equations (7), (8), and (9) form a simultaneous set of three equations in the four unknowns u_2, u_3, P_1, and T_3. Bernoulli's equation for the ambient/secondary airflow is used to close the set:

$$P_{amb} - P_1 = \frac{\rho_2u_2^2}{2} \quad (10)$$

since $P_1 = P_2$ and with $P_{amb} = P_3$; or

$$u_2 = \sqrt{\frac{2RT_2(P_3 - P_1)}{P_1}} \quad (11)$$

To facilitate the mathematical analysis, equations (7), (8), (9), and (11) are recast in the following form:

$$P_1(K_1 + K_2u_2) = K_3' \frac{u_3}{T_3} \tag{12}$$

$$P_1(K_1' + K_2u_2^2) = K_3 + K_3' \frac{u_3^2}{T_3} \tag{13}$$

$$P_1\left[K_4 + K_2u_2\left[K_5 + \frac{u_2^2}{2}\right]\right] = \frac{K_3'u_3}{T_3}\left[c_pT_3 + \frac{u_3^2}{2}\right] \tag{14}$$

$$u_2 = \sqrt{\left[\frac{K_6(P_3 - P_1)}{P_1}\right]} \tag{15}$$

where

$$K_1 = \frac{A_1u_1}{RT_1} \tag{16a}$$

$$K_2 = \frac{A_2}{RT_2} \tag{16b}$$

$$K_3 = P_3A_3 \tag{16c}$$

$$K_3' = \frac{K_3}{R} \tag{16d}$$

$$K_1' = K_1u_1 + A_3 \tag{16e}$$

$$K_4 = K_1\left[c_pT_1 + \frac{u_1^2}{2}\right] \tag{16f}$$

$$K_5 = c_p T_2 \tag{16g}$$

$$K_6 = 2RT_2 \tag{16h}$$

These equations form a nonlinear set of simultaneous equations which must be solved by some iterative means. The following technique is suggested:

1) guess P_1;

2) from equation (15) with $P_{amb} = P_3$ and T_2 known, find u_2:

$$u_2 = \sqrt{\left[\frac{K_6(P_3 - P_1)}{P_1}\right]} \tag{15}$$

3) from equation (12), find u_3/T_3:

$$\frac{u_3}{T_3} = \frac{P_1(K1 + K2u_2)}{K_3'} \tag{17}$$

4) from equation (13), find u_3:

$$u_3 = \frac{\left[\dfrac{P_1(K_1' + K_2 u_2^2) - K_3}{K_3'}\right]}{u_3/T_3} \tag{18}$$

5) find T_3 from u_3 and u_3/T_3:

$$T_3 = \frac{u_3}{(u_3/T_3)} \tag{19}$$

6) calculate P_1 from equation (14):

$$P_1 = \frac{\dfrac{K_3'u_3}{T_3}\left[c_pT_3 + \dfrac{u_3^2}{2}\right]}{K_4 + K_2u_2\left[K_5 + \dfrac{u_2^2}{2}\right]} \qquad (20)$$

7) check P_1 from equation (20) to ensure that it is equal to the guessed value; otherwise guess P_1 again (see Appendix B).

3.0 SAMPLE CALCULATION

Given a four inch diameter nozzle and a 21 inch diameter mixing tube, calculate the exit flow rate and temperature, the secondary flow rate, and outlet/primary mass flow and volume flow ratios if the primary airflow is 120 lbm/min at 500 °F. The parameters of the problem are:

D_1 = 4 in =101.6 mm, thus A_1 = 0.0081073 m^2

T_1 = 500°F = 533.15 K

m_1 = 120 lbm/min = 0.907 kg/s, thus u_1 = 168.94 m/s

D_3 = 21 in = 533.4 mm, thus A_3 = 0.22346 m^2

$A_2 = A_3 - A_1$ = 0.21535 m^2

P_3 = 101325 Pa

T_2 = 288.15 K

c_p = 1004.76 J/kgK (air at 15°C)

R = 287.074 J/kgK (air at 15°C)

$K_1 = 8.9489 \times 10^{-6}$

$K_2 = 2.6034 \times 10^{-6}$

K_3 = 22642

$K_3' = 78.871$

$K_1' = 0.22497$

$K_4 = 4.9215$

$K_5 = 289522$

$K_6 = 165440$

Using the program SEPP given in Appendix A, and the method of guessing P_1 after the first two guesses given in Appendix B, the results of the calculations are displayed in Table 1.

Table 1 Sample Calculation Results

Iter #	$(P_1)_{in}$ (Pa)	u_2 (m/s)	u_3 (m/s)	T_3 (K)	$(P_1)_{out}$ (Pa)	%Diff
1	101000	23.07	31.57	357.2	112196	-9.9
2	100900	26.40	30.70	309.0	98090	2.9
3	100920	25.76	30.82	316.9	100411	0.51
4	100924.5	25.62	30.85	318.7	100953	-0.028
5	100924.26	25.63	30.85	318.6	100924.0	0.00027
6	100924.26	25.63	30.85	318.6	100924.26	-4.x10-7

The exit flow rate at an exit velocity of 30.85 m/s is (with density $\rho_3 = 1.108$ kg/m^3 calculated knowing T_3 and P_3):

$$Q_3 = u_3A_3 = 6.8942 \text{ m}^3/\text{s} = 14610 \text{ cfm}$$

$$m_3 = \rho_3Q_3 = 7.6373 \text{ kg/s} = 1010 \text{ lbm/min}$$

The inflow is :

$$Q_1 = u_1A_1 = 1.3697 \text{ m}^3/\text{s} = 2900 \text{ cfm}$$

$$m_1 = 0.903 \text{ kg/s} = 120 \text{ lbm/min}$$

The ratios are $Q_3/Q_1 = 5.0$, and $m_3/m_1 = 8.4$

4.0 DISCUSSION

The control volume takes into account only the mixing volume, i.e. the volume in which the two streams interact , and as such, a "back pressure" on the system cannot be determined. This is due to insufficient information on the upstream geometry, i.e. nozzle, etc.

The analysis does not take into account the mixing tube length, hence friction, and other wall interactions, are not accounted for. Previous experience (ELLIN 1977, KAVALIS 1983, LEMKE et al. 1977, and MOSS 1977) indicates that the mixing tube should be at least three mixing tube diameters long to promote adequate mixing of the primary and secondary streams. However because the mixing tube-to-primary area ratio is so large (27.6), a longer mixing tube would probably be needed, since the value of $L/D_m = 3$ was determined from the geometries studied in the References, where area ratios were of the order of $A_m/A_p = 2$ to $A_m/A_p = 3$. If the mixing tube is not long enough, the outlet velocity and temperature profiles would be highly non-uniform as shown in Figure 2. Of course, the problem with a very long mixing tube would be the eventual non-negligible effects of friction and shear layer/wall interaction effect, which would increase the losses.

The analysis is a highly idealized scenario for this particular type of air flow (see the list of assumptions). The entrainment and final outflow of air, therefore, is the maximum attainable for these types of conditions. The actual airflow will be lower for a variety of reasons: Much of the induced airflow in this type of system is caused by turbulent shear interaction between the primary and secondary flows. Because the nozzle is small, the shear interaction "surface" will have a small area, and it is doubtful whether much secondary flow would be induced -- certainly not the 15:1 ratio as was hoped for by AMDU personnel (Cook 1987). More secondary air flow could be induced by using a "lobed" primary nozzle, or splitting the primary flow into four smaller nozzles (ELLIN 1977). These increase the shear interaction surface area. Also, a "standoff" $S = 0.5\ D_m$ is suggested by ELLIN (1977) to increase the entrainment in an actual experiment [see Figure 3].

Finally, only an experimental analysis will be able to ultimately verify whether the analysis is correct. A parametric analysis, with geometry as variable (diameters, lengths), will indicate trends and optimum conditions.

5.0 REFERENCES

COOK, B. 1987. AMDU, Personal Communication to P. J. Turyk, NRCC, September.

ELLIN, C.R. 1977. Model Test of Multiple Nozzle Exhaust Gas Eductor Systems for Gas Turbine Powered Ships. M.Sc. Thesis, Dept. of Mechanical Engineering, Naval Postgraduate School, Monterey, CA. (NTIS Microfiche AD-A044 182)

KAVALIS, A.E. 1983. Effect of Shroud Geometry on the Effectiveness of a Short Mixing Stack Gas Eductor Model. M.Sc. Thesis, Dept. of Mechanical Engineering, Naval Postgraduate School, Monterey, CA. (NTIS Microfiche AD-A132 173)

LEMKE, R.J., C.P. Staehli. 1977. Performance of Multiple Nozzle Eductor Systems with Several Geometric Configurations. M.Sc. Thesis, Dept. of Mechanical Engineering, Naval Postgraduate School, Monterey, CA. (NTIS Microfiche AD-A061 278)

MOSS, C.M. 1977. Effect of Several Geometric Parameters on the Performance of Mutiple Nozzle Eductor Systems. M.Sc. Thesis, Dept. of Mechanical Engineering, Naval Postgraduate School, Monterey, CA. (NTIS Microfiche AD-A047 186)



Figure 1 Ejector Pump



Figure 2 Non Uniform Velocity and Temperature Profiles



Figure 3 Definition of Standoff

Appendix A

Simple Ejector Pump Program (SEPP) Listing

```
C***********************************************************************
C*                                                                     *
C*  Simple Ejector Pump Program  (SEPP)                                *
C*                                                                     *
C*  Programmed by Peter Turyk, NRC Engine Laboratory                   *
C*  October, 1987                                                      *
C*                                                                     *
C***********************************************************************
      PROGRAM SEPP
      IMPLICIT REAL*8 (A-Z)
      INTEGER I,N
      COMMON/CONST/K1,K2,K3,K31,K11,K4,K5,K6,P3,CP
      DATA PI/3.141592653589793D0/
C
C...Read input data from file EJECTOR.DAT
C
      OPEN(UNIT=1,FILE='SEPP.IN',TYPE='OLD',DISPOSE='SAVE')
      OPEN(UNIT=6,FILE='SEPP.OUT',TYPE='OLD',DISPOSE='SAVE')
      READ(1,*)D1,T1,U1
      READ(1,*)T2,P3,D3
      READ(1,*)CP,R,N
      write(5,*)D1,T1,U1
      write(5,*)T2,P3,D3
      write(5,*)CP,R,N
C
C...Calculate equation constants
C
      A1=PI*(D1/2.D0)**2
      A3=PI*(D3/2.D0)**2
      A2=A3-A1
      K1=A1*U1/R/T1
      K2=A2/R/T2
      K3=P3*A3
      K31=K3/R
      K11=K1*U1+A3
      K4=K1*(CP*T1+U1*U1/2.D0)
      K5=CP*T2
      K6=2.D0*R*T2
C
C...First guess and calculation
C
      P1IN=P3-200.D0
      CALL CONSRV(P1IN, P1OUT,U2,U3,T3)
      DIFF=(P1IN-P1OUT)/P1IN*100.D0
      WRITE(5,100)
      WRITE(6,100)
      WRITE(5,101)P1IN,P1OUT,DIFF,U2,U3,T3
      WRITE(6,101)P1IN,P1OUT,DIFF,U2,U3,T3
      CALL SWITCH(P1IN,P1OUT,DUM, P1INO,P1OUTO)
C
C...Second guess and begin loop
C
      P1IN=P1INO-100.D0
      DO 1000 I=1,N
        CALL CONSRV(P1IN, P1OUT,U2,U3,T3)
        DIFF=(P1IN-P1OUT)/P1IN*100.D0
        WRITE(5,101)P1IN,P1OUT,DIFF,U2,U3,T3
        WRITE(6,101)P1IN,P1OUT,DIFF,U2,U3,T3
C
```

```
C...If calculation has converged, exit loop, otherwise make new guess
C
          IF(DABS(P1IN-P1OUT).LE.1.D-8)GO TO 10
          CALL GUESS(P1IN0,P1OUT0,P1IN,P1OUT, PNEW)
          CALL SWITCH(P1IN,P1OUT,PNEW, P1IN0,P1OUT0)
 1000 CONTINUE
      WRITE(5,102)I,P1IN,P1OUT
      WRITE(6,102)I,P1IN,P1OUT
C
C...Calculate volume and mass flows
C
   10 Q1=U1*A1
      Q3=U3*A3
      M1=P1OUT*K1
      M3=K31*U3/T3
      C=60.D0/.3048D0**3
      C1=2.204622622*60.D0
      Q1CFM=Q1*C
      Q3CFM=Q3*C
      M3LBM=M3*C1
      M1LBM=M1*C1
      QRAT=Q3/Q1
      MRAT=M3/M1
      WRITE(5,103)Q1,Q1CFM,Q3,Q3CFM,M1,M1LBM,M3,M3LBM,QRAT,MRAT
      WRITE(6,103)Q1,Q1CFM,Q3,Q3CFM,M1,M1LBM,M3,M3LBM,QRAT,MRAT
C
C...Format statements
C
  100 FORMAT(T15,'Simple Ejector Pump Program'/T15,27('-')///
     1       T10,'P1(guess)',T25,'P1(calc)',T41,'% Diff',
     2       T56,'U2',T66,'U3',T77,'T3'/T12,'(Pa)',T27,'(Pa)',
     3       T55,'(m/s)',T65,'(m/s)',T77,'(K)'//)
  101 FORMAT(T10,F9.2,T25,F9.2,T40,D8.1,T55,F5.2,T65,F5.2,T75,F6.2)
  102 FORMAT(T15,'*****WARNING*****'/T15,
     1       T15,'Maximum number of iterations has been reached:  ',I3/
     2       T15,'P1(guess) = ',F9.2,4X,'P1(calc) = ',F9.2/
     3       T15,'*****************')
  103 FORMAT(//T15,'Primary volume flow:',2X,F7.3,' (m3/s)',
     1         5X,F7.0,' (cfm)'/
     1         T15,'Exit volume flow   :',2X,F7.3,' (m3/s)',
     3         5X,F7.0,' (cfm)'/
     2         T15,'Primary mass flow  :',2X,F7.3,' (kg/s)',
     5         5X,F6.1,' (lbm/min)'/
     3         T15,'Exit mass flow     :',2X,F7.3,' (kg/s)',
     5         5X,F6.1,' (lbm/min)'/
     4         T15,'Volume flow ratio           :',T50,F5.1/
     5         T15,'Mass flow ratio             :',T50,F5.1)
      STOP 'Happy ejecting'
      END
C************************************
C*                                  *
C*  Subroutine C O N S R V          *
C*                                  *
C************************************
      SUBROUTINE CONSRV(P1, P1OUT,U2,U3,T3)
      IMPLICIT REAL*8 (A-Z)
      COMMON/CONST/K1,K2,K3,K31,K11,K4,K5,K6,P3,CP
      U2=DSQRT(K6*(P3-P1)/P1)
      U3DT3=P1*(K1+K2*U2)/K31
      U3=(P1*(K11+K2*U2*U2)-K3)/K31/U3DT3
```

```
      T3=U3/U3DT3
      P1OUT=K31*U3/T3*(CP*T3+U3*U3/2.D0)/(K4+K2*U2*(K5+U2*U2/2.D0))
      RETURN
      END
C*************************************
C*                                   *
C*  Subroutine G U E S S             *
C*                                   *
C*************************************
      SUBROUTINE GUESS(X0,Y0,X1,Y1, XNEW)
      IMPLICIT REAL*8 (A-Z)
      M=(Y1-Y0)/(X1-X0)
      XNEW=(M*X1-Y1)/(M-1.D0)
      RETURN
      END
C*************************************
C*                                   *
C*  Subroutine S W I T C H           *
C*                                   *
C*************************************
      SUBROUTINE SWITCH(X1,Y1,XNEW, X0,Y0)
      IMPLICIT REAL*8 (A-Z)
      Y0=Y1
      X0=X1
      X1=XNEW
      RETURN
      END
```

Appendix B

Iterative Interpolation/Extrapolation Technique to Determine New Inlet Pressure

In the ejector pump analysis, a non-linear set of equations must be solved given a guessed input of P_1. The set of equations are solved when the guessed P_1 equals the P_1 calculated in the equations.

The calculation starts when two guesses of P_1 are made with corresponding calculated P_1, and then a third guess of P_1 is made based on these two previous pairs, with the objective of attaining

$$(P_1)_{guess} = (P_1)_{calc} \qquad \text{(B1)}$$

or

$$P_{1g} = P_{1c} \qquad \text{(B1a)}$$

The situtlon can be generalized as follows: given two inputs x_1 and x_2, with corresponding outputs y_1 and y_2, where:

x_1 .ne. y_1,

x_2 .ne. y_2,

x_1 .ne. x_2,

It is desired to find x_3 such that $x_3 = y_3$ based on a linear interpolation between (x_1, y_1) and (x_2, y_2) pairs. This is done by finding the intersection of the line defined by (x_1, y_1) and (x_2, y_2), and the line $x = y$, as shown in Figure B1.

The intersection is found by solving a set of two simultaneous equations

$$y = x \qquad \text{(B2)}$$

$$y = mx + b \qquad \text{(B3)}$$

where

$$m = \frac{y_2 - y_1}{x_2 - x_1} \qquad \text{(B4)}$$

$$b = y_1 - mx_1. \qquad \text{(B5)}$$

Subtracting (B2) from (B3) gives

$$x_3 = \frac{-b}{m - 1}$$

$$x_3 = \frac{mx_1 - y_1}{m - 1} \qquad \text{(B6)}$$

To find the value of P_{1g} which will yield $P_{1g} = P_{1c}$, equation (B6) is used where

$$x = P_{1g} \qquad \text{(B7)}$$

$$y = P_{1c} \qquad \text{(B8)}$$

$$m = \frac{P_{1c}^{2} - P_{1c}^{1}}{P_{1g}^{2} - P_{1g}^{1}} \qquad \text{(B9)}$$

where the superscripts denote the existing two guesses. Once P_{1g}^3 is determined, the previous two guesses are updated as follows to ensure that the latest guesses are used to calculate the next guess:

$$P_{1c}^{2} \dashrightarrow P_{1c}^{1} \qquad \text{(B10)}$$

$$P_{1g}^{2} \dashrightarrow P_{1g}^{1} \qquad \text{(B11)}$$

$$P_{1g}^{3} \dashrightarrow P_{1g}^{2} \qquad \text{(B12)}$$

A new P_{1c} is calculated — this becomes P_{1c}^2 — and the process is repeated until $P_{1c} = P_{1g}$.



Figure B1

REPORT DOCUMENTATION PAGE/PAGE DE DOCUMENTATION DE RAPPORT

REPORT/RAPPORT	REPORT/RAPPORT
1a LM-ENG-014	1b
SECURITY CLASSIFICATION/ CLASSIFICATION DE SÉCURITÉ 2 Unclassified	DISTRIBUTION/DIFFUSION ☒ Controlled/Contrôlé ☐ Unlimited/Illimité 3

TITLE/SUBTITLE/TITRE/SOUS-TITRE

4 Simple Ejector Pump Analysis

AUTHOR(S)/AUTEUR(S)

5 P.J. Turyk

SERIES/SÉRIE

6

CORPORATE AUTHOR/PERFORMING AGENCY/AUTEUR D'ENTREPRISE/AGENCE D'EXÉCUTION

National Research Council Canada Engine Laboratory
7 Division of Mechanical Engineering

SPONSORING AGENCY/AGENCE DE SUBVENTION

8

DATE	FILE/DOSSIER	LAB. ORDER COMMANDE DU LAB.	PAGES	DIAGS	REFS
9 1988/08	10	11	14 (12a)	4 (12b)	5 (12c)

NOTES

13

DESCRIPTORS(KEY WORDS)/MOTS-CLÉS

Ejector, eductor, fluid mechanics, one-dimensional, mass momentum,
14 energy, numerical simulation

SUMMARY/SOMMAIRE

Present one-dimensional solution of conservation of mass, momentum, and energy for an ejector pump. Presents mathematical formulation and sample calculation. Includes a computer program.
15

ADDRESS/ADRESSE

D.M. Rudnitski
Section Head, Engine Laboratory
Division of Mechanical Engineering
National Research Council Canada
Bldg. M-7, Montreal Rd.
16 Ottawa, Ontario K1A 0R6